
Installation

If you have a running Julia distribution (Julia 0.4 or greater), you have everything you need to install POMDPs.jl. To install the package, simply
run the following from the Julia REPL:

Pkg.add("POMDPs") # installs the POMDPs.jl package





Once you have POMDPs.jl installed, you can install any package that is part of the JuliaPOMDP community by running:

using POMDPs
POMDPs.add("SARSOP") # installs the SARSOP solver





The code above will download and install all dependencies automatically. All JuliaPOMDP packages have been tested on
Linux and OS X, and most have been tested on Windows.

To get a list of all the available packages run:

POMDPs.available() # prints a list of all the available packages that can be installed with POMDPs.add





Due to the modular nature of the framework, you can choose to only install select solvers/support tools. However,
if you want to install all of the supported JuliaPOMDP packages you can run the following code:

POMDPs.add_all() # installs all the JuliaPOMDP packages (may take a few minutes)





If you want to avoid any non-Julia dependencies, run:

POMDPs.add_all(native_only=true)









          

      

      

    

  

    
      
          
            
  
[Generative POMDP Interface](@id generative_doc)


Description

The generative interface contains a small collection of functions that makes implementing and solving problems with generative models easier. These functions return states and observations instead of distributions as in the [Explicit interface](@ref explicit_doc).

The functions are:

generate_s(pomdp, s, a, rng) -> sp
generate_o(pomdp, s, a, sp, rng) -> o
generate_sr(pomdp, s, a, rng) -> (s, r)
generate_so(pomdp, s, a, rng) -> (s, o)
generate_or(pomdp, s, a, sp, rng) -> (o, r)
generate_sor(pomdp, s, a, rng) -> (s, o, r)
initial_state(pomdp, rng) -> s





Each generate_ function is a single step simulator that returns a new state, observation, reward, or a combination given the current state and action (and sp in some cases). rng is a random number generator such as Base.GLOBAL_RNG or another MersenneTwister that is passed as an argument and should be used to generate all random numbers within the function to ensure that all simulations are exactly repeatable. [http://docs.julialang.org/en/release-0.5/stdlib/numbers/#random-numbers]

The functions that do not deal with observations may be defined for MDPs as well as POMDPs.

A problem writer will generally only have to implement one or two of these functions for all solvers to work (see below).




Example

The following example shows an implementation of the Crying Baby problem [1]. A definition of this problem using the explicit interface is given in the POMDPModels package [https://github.com/JuliaPOMDP/POMDPModels.jl].

importall POMDPs

# state: true=hungry, action: true=feed, obs: true=crying

type BabyPOMDP <: POMDP{Bool, Bool, Bool}
    r_feed::Float64
    r_hungry::Float64
    p_become_hungry::Float64
    p_cry_when_hungry::Float64
    p_cry_when_not_hungry::Float64
    discount::Float64
end
BabyPOMDP() = BabyPOMDP(-5., -10., 0.1, 0.8, 0.1, 0.9)

discount(p::BabyPOMDP) = p.discount

function generate_s(p::BabyPOMDP, s::Bool, a::Bool, rng::AbstractRNG)
    if s # hungry
        return true
    else # not hungry
        return rand(rng) < p.p_become_hungry ? true : false
    end
end

function generate_o(p::BabyPOMDP, s::Bool, a::Bool, sp::Bool, rng::AbstractRNG)
    if sp # hungry
        return rand(rng) < p.p_cry_when_hungry ? true : false
    else # not hungry
        return rand(rng) < p.p_cry_when_not_hungry ? true : false
    end
end

# r_hungry
reward(p::BabyPOMDP, s::Bool, a::Bool) = (s ? p.r_hungry : 0.0) + (a ? p.r_feed : 0.0)

initial_state_distribution(p::BabyPOMDP) = [false] # note rand(rng, [false]) = false, so this is encoding that the baby always starts out full





This can be solved with the POMCP solver.

using POMCP
using POMDPToolbox

pomdp = BabyPOMDP()
solver = POMCPSolver()
planner = solve(solver, pomdp)

hist = simulate(HistoryRecorder(max_steps=10), pomdp, planner);
println("reward: $(discounted_reward(hist))")








Which function(s) should I implement for my problem / use in my solver?


Problem Writers

Generally, a problem implementer need only implement the simplest one or two of these functions, and the rest are automatically synthesized at runtime.

If there is a convenient way for the problem to generate a combination of states, observations, and rewards simultaneously (for example, if there is a simulator written in another programming language that generates these from the same function, or if it is computationally convenient to generate sp and o simultaneously), then the problem writer may wish to directly implement one of the combination generate_ functions, e.g. generate_sor() directly.

Use the following logic to determine which functions to implement:


	If you are implementing the problem from scratch in Julia, implement generate_s and generate_o.

	Otherwise, if your external simulator returns x, where x is one of sr, so, or, or sor, implement generate_x. (you may also have to implement generate_s separately for use in particle filters).



Note: if an explicit definition is already implemented, you do not need to implement any functions from the generative interface - POMDPs.jl will automatically generate implementations of them for you at runtime (see generative_impl.jl).




Solver and Simulator Writers

Solver writers should use the single function that generates everything that they need and nothing they don’t. For example, if the solver needs access to the state, observation, and reward at every timestep, they should use generate_sor() rather than generate_s() and generate_or(), and if the solver needs access to the state and reward, they should use generate_sr() rather than generate_sor(). This will ensure the widest interoperability between solvers and problems.

In other words, if you need access to x where x is s, o, sr, so, or, or sor at a certain point in your code, use generate_x.

[1] Decision Making Under Uncertainty: Theory and Application by
Mykel J. Kochenderfer, MIT Press, 2015









          

      

      

    

  

    
      
          
            
  
Defining a Solver

In this section, we will walk through an implementation of the
QMDP [http://www-anw.cs.umass.edu/~barto/courses/cs687/Cassandra-etal-POMDP.pdf] algorithm. QMDP is the fully
observable approximation of a POMDP policy, and relies on the Q-values to determine actions.


Background

Let’s say we are working with a POMDP defined by the tuple $(\mathcal{S}, \mathcal{A}, \mathcal{Z}, T, R, O, \gamma)$,
where $\mathcal{S}$, $\mathcal{A}$, $\mathcal{Z}$ are the discrete state, action, and observation spaces
respectively. The QMDP algorithm assumes it is given a discrete POMDP. In our model $T : \mathcal{S} \times
\mathcal{A} \times \mathcal{S} \rightarrow [0, 1]$ is the transition function, $R: \mathcal{S} \times \mathcal{A}
\rightarrow \mathbb{R}$ is the reward function, and $O: \mathcal{Z} \times \mathcal{A} \times \mathcal{S} \rightarrow
[0,1]$ is the observation function. In a POMDP, our goal is to compute a policy $\pi$ that maps beliefs to actions $\pi: b \rightarrow a$. For
QMDP, a belief can be represented by a discrete probability distribution over the state space (although there may be
other ways to define a belief in general and POMDPs.jl allows this flexibility).

It can be shown (e.g. in [1], section 6.3.2) that the optimal value function for a POMDP can be written in terms of alpha vectors. In the QMDP approximation, there is a single alpha vector that corresponds to each action ($\alpha_a$), and the policy is computed according to

$\pi(b) = \underset{a}{\text{argmax}} , \alpha_{a}^{T}b$

Thus, the alpha vectors can be used to compactly represent a QMDP policy.




QMDP Algorithm

QMDP uses the columns of the Q-matrix obtained by solving the MDP defined by $(\mathcal{S}, \mathcal{A}, T, R, \gamma)$ (that is, the fully observable MDP that forms the basis for the POMDP we are trying to solve).
If you are familiar with the value iteration algorithm for MDPs, the procedure for finding these alpha vectors is identical. Let’s first
initialize the alpha vectors $\alpha_{a}^{0} = 0$ for all $s$, and then iterate

$\alpha_{a}^{k+1}(s) = R(s,a) + \gamma \sum_{s’} T(s’|s,a) \max_{a’} \alpha_{a’}^{k}(s’)$

After enough iterations, the alpha vectors converge to the QMDP approximation.

Remember that QMDP is just an approximation method, and does not guarantee that the alpha vectors you obtain actually
represent your POMDP value function. Specifically, QMDP has trouble in problems with information gathering actions
(because we completely ignored the observation function when computing our policy). However, QMDP works very well in problems where a particular choice of action has
little impact on the reduction in state uncertainty.




Requirements for a Solver

Before getting into the implementation details, let’s first go through what a POMDP solver must be able to do and support. We need three custom types that inherit from abstract types in POMDPs.jl. These type are Solver, Policy, and Updater. It is usually useful to have a custom type that represents the belief used by your policy as well.

The requirements are as follows:

# types
QMDPSolver
QMDPPolicy
DiscreteUpdater # already implemented for us in POMDPToolbox
DiscreteBelief # already implemented for us in POMDPToolbox
# methods
updater(p::QMDPPolicy) # returns a belief updater suitable for use with QMDPPolicy
initialize_belief(bu::DiscreteUpdater, initial_state_dist) # returns a Discrete belief
solve(solver::QMDPSolver, pomdp::POMDP) # solves the POMDP and returns a policy
update(bu::DiscreteUpdater, belief_old::DiscreteBelief, action, obs) # returns an updated belied (already implemented)
action(policy::QMDPPolicy, b::DiscreteBelief) # returns a QMDP action





You can find the implementations of these types and methods below.




Defining the Solver and Policy Types

Let’s first define the Solver type. The QMDP solver type should contain all the information needed to compute a policy (other than the problem itself). This information can be thought of as the hyperparameters of the solver. In QMDP, we only need two hyper-parameters. We may want to set the maximum number of iterations that the algorithm runs for, and a tolerance value (also known as the Bellman residual). Both of these quantities define terminating criteria for the algorithm. The algorithm stops either when the maximum number of iterations has been reached or when the infinity norm of the difference in utility values between two iterations goes below the tolerance value. The type definition has the form:

using POMDPs # first load the POMDPs module
type QMDPSolver <: Solver
    max_iterations::Int64 # max number of iterations QMDP runs for
    tolerance::Float64 # Bellman residual: terminates when max||Ut-Ut-1|| < tolerance
end
# default constructor
QMDPSolver(;max_iterations::Int64=100, tolerance::Float64=1e-3) = QMDPSolver(max_iterations, tolerance)





Note that the QMDPSolver inherits from the abstract Solver type that’s part of POMDPs.jl.

Now, let’s define a policy type. In general, the policy should contain all the information needed to map a belief to an action. As mentioned earlier, we need alpha vectors to be part of our policy. We can represent the alpha vectors using a matrix of size $|\mathcal{S}| \times |\mathcal{A}|$. Recall that in POMDPs.jl, the actions can be represented in a number of ways (Int64, concrete types, etc), so we need a way to map these actions to integers so we can index into our alpha matrix. The type looks like:

using POMDPToolbox # for ordered_actions

type QMDPPolicy <: Policy
    alphas::Matrix{Float64} # matrix of alpha vectors |S|x|A|
    action_map::Vector{Any} # maps indices to actions
    pomdp::POMDP            # models for convenience
end
# default constructor
function QMDPPolicy(pomdp::POMDP)
    ns = n_states(pomdp)
    na = n_actions(pomdp)
    alphas = zeros(ns, na)
    am = Any[]
    space = ordered_actions(pomdp)
    for a in iterator(space)
        push!(am, a)
    end
    return QMDPPolicy(alphas, am, pomdp)
end





Now that we have our solver and policy types, we can write the solve function to compute the policy.




Writing the Solve Function

The solve function takes in a solver, a POMDP, and an optional policy argument. Let’s compute those alpha vectors!

function POMDPs.solve(solver::QMDPSolver, pomdp::POMDP)

    policy = QMDPPolicy(pomdp)

    # get solver parameters
    max_iterations = solver.max_iterations
    tolerance = solver.tolerance
    discount_factor = discount(pomdp)

    # intialize the alpha-vectors
    alphas = policy.alphas

    # initalize space
    sspace = ordered_states(pomdp)  # returns a discrete state space object of the pomdp
    aspace = ordered_actions(pomdp) # returns a discrete action space object

    # main loop
    for i = 1:max_iterations
        residual = 0.0
        # state loop
        # the iterator function returns an iterable object (array, iterator, etc) over a discrete space
        for (istate, s) in enumerate(sspace)
            old_alpha = maximum(alphas[istate,:]) # for residual
            max_alpha = -Inf
            # action loop
            # alpha(s) = R(s,a) + discount_factor * sum(T(s'|s,a)max(alpha(s'))
            for (iaction, a) in enumerate(aspace)
                # the transition function modifies the dist argument to a distribution availible from that state-action pair
                dist = transition(pomdp, s, a) # fills distribution over neighbors
                q_new = 0.0
                for sp in iterator(dist)
                    # pdf returns the probability mass of sp in dist
                    p = pdf(dist, sp)
                    p == 0.0 ? continue : nothing # skip if zero prob
                    # returns the reward from s-a-sp triple
                    r = reward(pomdp, s, a, sp)
    
                    # state_index returns an integer
                    sidx = state_index(pomdp, sp)
                    q_new += p * (r + discount_factor * maximum(alphas[sidx,:]))
                end
                new_alpha = q_new
                alphas[istate, iaction] = new_alpha
                new_alpha > max_alpha ? (max_alpha = new_alpha) : nothing
            end # actiom
            # update the value array
            diff = abs(max_alpha - old_alpha)
            diff > residual ? (residual = diff) : nothing
        end # state
        # check if below Bellman residual
        residual < tolerance ? break : nothing
    end # main
    # return the policy
    policy
end





At each iteration, the algorithm iterates over the state space and computes an alpha vector for each action. There is a check at the end to see if the Bellman residual has been satisfied. The solve function assumes the following POMDPs.jl functions are implemented by the user of QMDP:

states(pomdp) # (in ordered_states) returns a state space object of the pomdp
actions(pomdp) # (in ordered_actions) returns the action space object of the pomdp
iterator(space) # returns an iterable object (array or iterator), used for discrete spaces only
transition(pomdp, s, a) # returns the transition distribution for the s, a pair
reward(pomdp, s, a, sp) # returns real valued reward from s, a, sp triple
pdf(dist, sp) # returns the probability of sp being in dist
state_index(pomdp, sp) # returns the integer index of sp (for discrete state spaces)





Now that we have a solve function, we define the action function to let users evaluate the policy:

function POMDPs.action(policy::QMDPPolicy, b::DiscreteBelief)
    alphas = policy.alphas
    ihi = 0
    vhi = -Inf
    (ns, na) = size(alphas)
    @assert length(b.b) == ns "Length of belief and alpha-vector size mismatch"
    # see which action gives the highest util value
    for ai = 1:na
        util = dot(alphas[:,ai], b.b)
        if util > vhi
            vhi = util
            ihi = ai
        end
    end
    # map the index to action
    return policy.action_map[ihi]
end








Belief Updates

Let’s now talk about how we deal with beliefs. Since QMDP is a discrete POMDP solver, we can assume that the user will represent their belief as a probablity distribution over states. That means that we can also use a discrete belief to work with our policy!
Lucky for us, the JuliaPOMDP organization contains tools that we can use out of the box for working with discrete beliefs. The POMDPToolbox package contains a DiscreteBelief type that does exactly what we need. The updater function allows us to declare that the DiscreteUpdater is the default updater to be used with a QMDP policy:

using POMDPToolbox # remeber to load the package that implements discrete beliefs for us
POMDPs.updater(p::QMDPPolicy) = DiscreteUpdater(p.pomdp) 





These are all the functions that you’ll need to have a working POMDPs.jl solver. Let’s now use existing benchmark models to evaluate it.




Evaluating the Solver

We’ll use the POMDPModels package from JuliaPOMDP to initialize a Tiger POMDP problem and solve it with QMDP.

using POMDPModels

# initialize model and solver
pomdp = TigerPOMDP()
solver = QMDPSolver()

# compute the QMDP policy
policy = solve(solver, pomdp)

# initalize updater and belief
b_up = updater(policy)
init_dist = initial_state_distribution(pomdp)

# create a simulator object for recording histories
sim_hist = HistoryRecorder(max_steps=100)

# run a simulation
r = simulate(sim_hist, pomdp, policy, b_up, init_dist)





That’s all you need to define a solver and evaluate its performance!




Defining Requirements

If you share your solver, in order to make it easy to use, specifying requirements as described [here](@ref specifying_requirements) is highly recommended.

[1] Decision Making Under Uncertainty: Theory and Application by
Mykel J. Kochenderfer, MIT Press, 2015







          

      

      

    

  

    
      
          
            
  
Getting Started

Before writing our own POMDP problems or solvers, let’s try out some of the available solvers and problem models
available in JuliaPOMDP.

Here is a short piece of code that solves the Tiger POMDP using SARSOP, and evaluates the results. Note that you must
have the SARSOP, POMDPModels, and POMDPToolbox modules installed.

using SARSOP, POMDPModels, POMDPToolbox

# initialize problem and solver
pomdp = TigerPOMDP() # from POMDPModels
solver = SARSOPSolver() # from SARSOP

# compute a policy
policy = solve(solver, pomdp)

#evaluate the policy
belief_updater = updater(policy) # the default QMPD belief updater (discrete Bayesian filter)
init_dist = initial_state_distribution(pomdp) # from POMDPModels
hr = HistoryRecorder(max_steps=100) # from POMDPToolbox
hist = simulate(hr, pomdp, policy, belief_updater, init_dist) # run 100 step simulation
println("reward: $(discounted_reward(hist))")





The first part of the code loads the desired packages and initializes the problem and the solver. Next, we compute a
POMDP policy. Lastly, we evaluate the results.

There are a few things to mention here. First, the TigerPOMDP type implements all the functions required by
SARSOPSolver to compute a policy. Second, each policy has a default updater (essentially a filter used to update the
belief of the POMDP). To learn more about Updaters check out the Concepts [http://juliapomdp.github.io/POMDPs.jl/latest/concepts/] section.





          

      

      

    

  

    
      
          
            
  
[Specifying Requirements](@id specifying_requirements)


Purpose

When a researcher or student wants to use a solver in the POMDPs ecosystem, the first question they will ask is “What do I have to implement to use this solver?”. The requirements interface provides a standard way for solver writers to answer this question.




Internal interface

The most important functions in the requirements interface are get_requirements, check_requirements, and show_requirements.

get_requirements(f::Function, args::Tuple{...}) should be implemented by a solver or simulator writer for all important functions that use the POMDPs.jl interface. In practice, this function will rarely by implemented directly because the [@POMDP_require](@ref pomdp_require_section) macro automatically creates it. The function should return a RequirementSet object containing all of the methods POMDPs.jl functions that need to be implemented for the function to work with the specified arguments.

check_requirements returns true if [all of the requirements in a RequirementSet are met](@ref implemented_section), and show_requirements prints out a list of the requirements in a RequirementSet and indicates which ones have been met.




[@POMDP_require](@id pomdp_require_section)

The @POMDP_require macro is the main point of interaction with the requirements system for solver writers. It uses a special syntax to automatically implement get_requirements. This is best shown by example. Consider this @POMDP_require block from the DiscreteValueIteration package [https://github.com/JuliaPOMDP/DiscreteValueIteration.jl]:

@POMDP_require solve(solver::ValueIterationSolver, mdp::Union{MDP,POMDP}) begin
    P = typeof(mdp)
    S = state_type(P)
    A = action_type(P)
    @req discount(::P)
    @req n_states(::P)
    @req n_actions(::P)
    @subreq ordered_states(mdp)
    @subreq ordered_actions(mdp)
    @req transition(::P,::S,::A)
    @req reward(::P,::S,::A,::S)
    @req state_index(::P,::S)
    as = actions(mdp)
    ss = states(mdp)
    @req iterator(::typeof(as))
    @req iterator(::typeof(ss))
    s = first(iterator(ss))
    a = first(iterator(as))
    dist = transition(mdp, s, a)
    D = typeof(dist)
    @req iterator(::D)
    @req pdf(::D,::S)
end





The first expression argument to the macro is a function signature specifying what the requirements apply to. The above example implements get_requirements{P<:Union{POMDP,MDP}}(solve::typeof(solve), args::Tuple{ValueIterationSolver,P}) which will construct a RequirementSet containing the requirements for executing the solve function with ValueIterationSolver and MDP or POMDP arguments at runtime.

The second expression is a begin-end block [http://docs.julialang.org/en/release-0.5/manual/control-flow/#compound-expressions] that specifies the requirements. The arguments in the function signature (solver and mdp in this example) may be used within the block.

The @req macro is used to specify a required function. Each @req should be followed by a function with the argument types specified. The @subreq macro is used to denote that the requirements of another function are also required. Each @subreq should be followed by a function call.




requirements_info

While the @POMDP_require macro is used to specify requirements for a specific method, the requirements_info function is a more flexible communication tool for a solver writer. requirements_info should print out a message describing the requirements for a solver. The exact form of the message is up to the solver writer, but it should be carefully thought-out because problem-writers will be directed to call the function (via the @requirements_info macro) as the first step in using a new solver (see tutorial).

By default, requirements_info calls show_requirements on the solve function. This is adequate in many cases, but in some cases, notably for online solvers such as MCTS [https://github.com/JuliaPOMDP/MCTS.jl], the requirements for solve do not give a good indication of the requirements for using the solver. Instead, the requirements for action should be displayed. The following example shows a more informative version of requirements_info from the MCTS package. Since action requires a state argument, requirements_info prompts the user to provide one.

function POMDPs.requirements_info(solver::AbstractMCTSSolver, problem::Union{POMDP,MDP})
    if state_type(typeof(problem)) <: Number
        s = one(state_type(typeof(problem)))
        requirements_info(solver, problem, s)
    else
        println("""
            Since MCTS is an online solver, most of the computation occurs in `action(policy, state)`. In order to view the requirements for this function, please, supply a state as the third argument to `requirements_info`, e.g.

                @requirements_info $(typeof(solver))() $(typeof(problem))() $(state_type(typeof(problem)))()

                """)
    end
end

function POMDPs.requirements_info(solver::AbstractMCTSSolver, problem::Union{POMDP,MDP}, s)
    policy = solve(solver, problem)
    requirements_info(policy, s)
end

function POMDPs.requirements_info(policy::AbstractMCTSPolicy, s)
    @show_requirements action(policy, s)
end








@warn_requirements

The @warn_requirements macro is a useful tool to improve usability of a solver. It will show a requirements list only if some requirements are not met. It might be used, for example, in the solve function to give a problem writer a useful error if some required methods are missing (assuming the solver writer has already used @POMDP_require to specify the requirements for solve):

function solve(solver::ValueIterationSolver, mdp::Union{POMDP, MDP})
    @warn_requirements solve(solver, mdp)

    # do the work of solving
end





@warn_requirements does perform a runtime check of requirements every time it is called, so it should not be used in code that may be used in fast, high-performance loops.




[Determining whether a function is implemented](@id implemented_section)

When checking requirements in check_requirements, or printing in show_requirements, the implemented function is used to determine whether an implementation for a function is available. For example implemented(discount, Tuple{NewPOMDP}) should return true if the writer of the NewPOMDP problem has implemented discount for their problem. In most cases, the default implementation,

implemented(f::Function, TT::TupleType) = method_exists(f, TT)





will automatically handle this, but there may be cases in which you want to override the behavior of implemented, for example, if the function can be synthesized from other functions. Examples of this can be found in the default implementations of the generative interface funcitons [https://github.com/JuliaPOMDP/POMDPs.jl/blob/master/src/generative_impl.jl.jl].







          

      

      

    

  

    
      
          
            
  
API Documentation

Documentation for the POMDPs.jl user interface. You can get help for any type or
function in the module by typing ? in the Julia REPL followed by the name of
type or function. For example:

julia> using POMDPs
julia> ?
help?> reward
search: reward

  reward{S,A,O}(pomdp::POMDP{S,A,O}, state::S, action::A, statep::S)

  Returns the immediate reward for the s-a-s triple

  reward{S,A,O}(pomdp::POMDP{S,A,O}, state::S, action::A)

  Returns the immediate reward for the s-a pair

Spaces and Distributions

Two important components of the definitions of MDPs and POMDPs are spaces, which specify the possible states, actions, and observations in a problem and distributions, which define probability distributions. In order to provide for maximum flexibility spaces and distributions may be of any type (i.e. there are no abstract base types). Solvers and simulators will interact with space and distribution types using the functions defined below.


Spaces

A space object should contain the information needed to define the set of all possible states, actions or observations. The implementation will depend on the attributes of the elements. For example, if the space is continuous, the space object may only contain the limits of the continuous range. In the case of a discrete problem, a vector containing all states is appropriate for representing a state.

The following functions may be called on a space object:


	rand

	dimensions

	iterator

	sampletype






Distributions

A distribution object represents a probability distribution. The following functions may be called on a distribution object


	rand

	iterator

	sampletype

	pdf

	mode

	mean









          

      

      

    

  

    
      
          
            
  
[Interface Requirements for Problems](@id requirements)

Due to the large variety of problems that can be expressed as MDPs and POMDPs and the wide variety of solution techniques available, there is considerable variation in which of the POMDPs.jl interface functions must be implemented to use each solver. No solver requires all of the functions in the interface, so it is wise to determine which functions are needed before jumping into implementation.

Solvers can communicate these requirements through the @requirements_info and @show_requirements macros. @requirements_info should give an overview of the requirements for a solver, which is supplied as the first argument, the macro can usually be more informative if a problem is specified as the second arg. For example, if you are implementing a new problem NewMDP and want to use the DiscreteValueIteration solver, you might run the following:

[image: requirements_info for a new problem]

Note that a few of the requirements could not be shown because actions is not implemented for the new problem.

If you would like to see a list of all of the requirements for a solver, try running @requirements_info with a fully implemented model from POMDPModels, for example,

[image: requirements_info for a fully-implemented problem]

@show_requirements is a lower-level tool that can be used to show the requirements for a specific function call, for example

@show_requirements solve(ValueIterationSolver(), NewMDP())





or

policy = solve(ValueIterationSolver(), GridWorld())
@show_requirements action(policy, GridWorldState(1,1))





In some cases, a solver writer may not have specified the requirements, in which case the requirements query macros will output

[No requirements specified]





In this case, please file an issue on the solver’s github page to encourage the solver writer to specify requirements.





          

      

      

    

  

    
      
          
            
  
[Explicit POMDP Interface](@id explicit_doc)

When using the explicit interface, the transition and observation probabilities must be explicitly defined. This section gives examples of two ways to define a discrete POMDP that is widely used in the literature.

Note that there is no requirement that a problem defined using the explicit interface be discrete; it is equally easy to define a continuous problem using the explicit interface.


Functional Form Explicit POMDP

In this example we show how to implement the famous Tiger Problem [https://www.cs.rutgers.edu/~mlittman/papers/aij98-pomdp.pdf].

In this implementation of the problem we will assume that the agent get a reward of -1 for listening at the door,
a reward of -100 for encountering the tiger, and a reward of 10 for escaping. The probability of hearing the tiger
when listing at the tiger’s door is 85%, and the discount factor is a parameter in the TigerPOMDP object.

We define the Tiger POMDP type:

importall POMDPs
type TigerPOMDP <: POMDP{Bool, Int64, Bool}
    discount_factor::Float64
end
TigerPOMDP() = TigerPOMDP(0.95) # default contructor
discount(pomdp::TigerPOMDP) = pomdp.discount_factor





Notice that the TigerPOMDP inherits from the abstract POMDP type provided by POMDPs.jl.
Our type is defined TigerPOMDP <: POMDP{Bool, Int64, Bool}, indicating that our states are Bools, actions are Int64, and observations are Bool.
In our problem there are only two states (whether the tiger is behind the left or right door), three actions (go left, go right, and listen), and two observations (hear the tiger behind the left or right door). We thus use booleans for the states and observations, and integers for the actions.
Note that states, actions, and observations can use arrays, strings, complex data structures, or even custom types.

Suppose that once implemented, we want to solve Tiger problems using the QMDP solver.
To see what functions QMDP needs us to implement, use the @requirements_info macro (see [Interface Requirements for Problems](@ref requirements)).

POMDPs.add("QMDP")
using QMDP
@requirements_info QMDPSolver() TigerPOMDP() 





We will begin by implementing the state, action, and observation spaces and functions for initializing them and sampling from them.

# STATE SPACE
const TIGER_ON_LEFT = true
const TIGER_ON_RIGHT = false

states(pomdp::TigerPOMDP) = [TIGER_ON_LEFT, TIGER_ON_RIGHT]
n_states(pomdp::TigerPOMDP) = 2

# ACTION SPACE
const OPEN_LEFT = 0
const OPEN_RIGHT = 1
const LISTEN = 2

actions(pomdp::TigerPOMDP) = [OPEN_LEFT,OPEN_RIGHT,LISTEN]
n_actions(pomdp::TigerPOMDP) = 3
action_index(::TigerPOMDP, a::Int64) = a+1

# OBSERVATION SPACE
const OBSERVE_LEFT = true
const OBSERVE_RIGHT = false

observations(::TigerPOMDP) = [OBSERVE_LEFT, OBSERVE_RIGHT]
n_observations(::TigerPOMDP) = 2





Before we can implement the core transition, reward, and observation functions we need to define how distributions over states and observations work for the Tiger POMDP.
We need to sample from these distributions and compute their likelihoods.
Are states and observations are binary, so we can use Bernoulli distributions:

type TigerDistribution
    p_true::Float64
end
TigerDistribution() = TigerDistribution(0.5) # default constructor
iterator(d::TigerDistribution) = [true, false]

# returns the probability mass for discrete distributions
function pdf(d::TigerDistribution, v::Bool)
    if v
        return d.p_true
    else
        return 1 - d.p_true
    end
end

# sample from the distribution
rand(rng::AbstractRNG, d::TigerDistribution) = rand(rng) ≤ d.p_true





We can now define our transition, observation, and reward functions.
Transition and observation return the distribution over the next state and observation, and reward returns the scalar reward.

function transition(pomdp::TigerPOMDP, s::Bool, a::Int64)
    d = TigerDistribution()
    if a == OPEN_LEFT || a == OPEN_RIGHT
        d.p_true = 0.5 # reset the tiger's location, which is what QMDP wants
    elseif s == TIGER_ON_LEFT
        d.p_true = 1.0 # tiger is on left
    else
        d.p_true = 0.0  # tiger is on right
    end
    d
end

function observation(pomdp::TigerPOMDP, a::Int64, sp::Bool)
    d = TigerDistribution()
    # obtain correct observation 85% of the time
    if a == LISTEN
        d.p_true = sp == TIGER_ON_LEFT ? 0.85 : 0.15
    else
        d.p_true = 0.5 # reset the observation - we did not listen
    end
    d
end
observation(pomdp::TigerPOMDP, s::Bool, a::Int64, sp::Bool) = observation(pomdp, a, sp) # convenience function

function reward(pomdp::TigerPOMDP, s::Bool, a::Int64)
    # rewarded for escaping, penalized for listening and getting caught
    r = 0.0
    if a == LISTEN
        r -= 1.0 # action penalty
    elseif (a == OPEN_LEFT && s == TIGER_ON_LEFT) ||
           (a == OPEN_RIGHT && s == TIGER_ON_RIGHT)
        r -= 100.0 # eaten by tiger
    else
        r += 10.0 # opened the correct door
    end
    r
end
reward(pomdp::TigerPOMDP, s::Bool, a::Int64, sp::Bool) = reward(pomdp, s, a) # convenience function





The last thing we need for the Tiger POMDP is an initial distribution over the state space.
In POMDPs.jl we make a strong distinction between this distribution and a belief.
In most literature these two concepts are considered the same. However, in more general terms, a belief is something that is mapped to an action using a POMDP policy.
If the policy is represented as something other than alpha-vectors (a policy graph, tree, or a recurrent neural network to give a few examples), it
may not make sense to think of a belief as a probability distribution over the state space.
Thus, in POMDPs.jl we abstract the concept of a belief beyond a probability distribution, allowing users to use what makes the most sense.

In order to reconcile this difference, each policy has a function called initialize_belief which takes in an
initial state distriubtion and a policy, and converts the
distribution into what we call a belief in POMDPs.jl. As the problem writer we must provide initial_state_distribution:

initial_state_distribution(pomdp::TigerPOMDP) = TigerDistribution(0.5)





We have fully defined the Tiger POMDP.
We can use now use JuliaPOMDP solvers to compute and evaluate a policy:

using QMDP, POMDPToolbox

pomdp = TigerPOMDP()
solver = QMDPSolver()
policy = solve(solver, pomdp)

init_dist = initial_state_distribution(pomdp)
hist = HistoryRecorder(max_steps=100) # from POMDPToolbox
r = simulate(hist, pomdp, policy) # run 100 step simulation





Please note that you do not need to define all the functions for most solvers.
If you want to use a specific solver, you usually only need a subset of what is above. Notably, when the problem only requires a generative model, you need not define any distributions. See Interface Requirements for Problems.




Tabular Form Explicit POMDP

The DiscretePOMDP problem representation allows you to specify discrete POMDP problems in tabular form.
If you can write the transition probabilities, observation probabilities, and rewards in matrix form, you can use the DiscreteMDP or
DiscretePOMDP types from POMDPModels which automatically implements all required functionality.
Let us do this with the Tiger POMDP:

using POMDPModels

# write out the matrix forms

# REWARDS
R = [-1. -100 10; -1 10 -100] # |S|x|A| state-action pair rewards

# TRANSITIONS
T = zeros(2,3,2) # |S|x|A|x|S|, T[s', a, s] = p(s'|a,s)
T[:,:,1] = [1. 0.5 0.5; 0 0.5 0.5]
T[:,:,2] = [0. 0.5 0.5; 1 0.5 0.5]

# OBSERVATIONS
O = zeros(2,3,2) # |O|x|A|x|S|, O[o, a, s] = p(o|a,s)
O[:,:,1] = [0.85 0.5 0.5; 0.15 0.5 0.5]
O[:,:,2] = [0.15 0.5 0.5; 0.85 0.5 0.5]

discount = 0.95
pomdp = DiscretePOMDP(T, R, O, discount)

# solve the POMDP the same way
solver = QMDPSolver()
policy = solve(solver, pomdp)





It is often easiest to define smaller problems in tabular form. However, for larger problems it can be
tedious and the functional form may be preferred. You can usually use any supported POMDP solver to solve these types of problems (the performance of the policy may vary however - SARSOP will usually outperform QMDP).







          

      

      

    

  

    
      
          
            
  
Frequently Asked Questions (FAQ)


Why am I getting a “No implementation for ...” error?

You will typically see this error when you haven’t implemented a function that a solver is trying to call.
For example, if you are using the QMDP solver, and have not implemented num_states for your POMDP, you will see the no
implementation error. To fix the error, you need to create a num_states function that takes in your POMDP. To see the
required functions for a given solver you can run:

using QMDP
QMDP.required_methods()








How do I save my policies?

We recommend using JLD [https://github.com/JuliaIO/JLD.jl] to save the whole policy object. This is a simple and
fairly efficient way to save Julia objects. JLD uses the HDF5 format underneath. To save a computed policy, run:

using JLD
save("my_policy.jld", "policy", policy)








Why isn’t the solver working?

There could be a number of things that are going wrong. Remeber, POMDPs can be failry hard to work with, but don’t
panic.
If you have a discrete POMDP or MDP and you’re using a solver that requires the explicit transition probabilities
(you’ve implemented a pdf function), the first thing to try is make sure that your probability masses sum up to unity.
We’ve provide some tools in POMDPToolbox that can check this for you.
If you have a POMDP called pomdp, you can run the checks by doing the following:

using POMDPToolbox
probability_check(pomdp) # checks that both observation and transition functions give probs that sum to unity
obs_prob_consistency_check(pomdp) # checks the observation probabilities
trans_prob_consistency_check(pomdp) # check the transition probabilities





If these throw an error, you may need to fix your transition or observation functions.




Why do I need to put type assertions pomdp::POMDP into the function signature?

Specifying the type in your function signature allows Julia to call the appropriate function when your custom type is
passed into it.
For example if a POMDPs.jl solver calls states on the POMDP that you passed into it, the correct states function
will only get dispatched if you specified that the states function you wrote works with your POMDP type. Because Julia
supports multiple-dispatch, these type assertion are a way for doing object-oriented programming in Julia.




Why are all the solvers in seperate modules?

We did not put all the solvers and support tools into POMDPs.jl, because we wanted POMDPs.jl to be a lightweight
interface package.
This has a number of advantages. The first is that if a user only wants to use a few solvers from the
JuliaPOMDP organization, they do not have to install all the other solvers and their dependencies.
The second advantage is that people who are not directly part of the JuliaPOMDP organization can write their own solvers
without going into the source code of other solvers. This makes the framework easier to adopt and to extend.







          

      

      

    

  

    
      
          
            
  
POMDPs

A Julia interface for defining, solving and simulating partially observable Markov decision processes and their fully
observable counterparts.


Package Features


	General interface that can handle problems with discrete and continuous state/action/observation spaces

	A number of popular state-of-the-art solvers available to use out of the box

	Tools that make it easy to define problems and simulate solutions

	Simple integration of custom solvers into the existing interface






Available Packages

The POMDPs.jl package contains the interface used for expressing and solving Markov decision processes (MDPs) and partially
observable Markov decision processes (POMDPs) in the Julia programming language. The
JuliaPOMDP [https://github.com/JuliaPOMDP] community maintains these packages. The packages currently maintained by
JuliaPOMDP are as follows:


MDP solvers:


	Value Iteration [https://github.com/JuliaPOMDP/DiscreteValueIteration.jl]

	Monte Carlo Tree Search [https://github.com/JuliaPOMDP/MCTS.jl]






POMDP solvers:


	QMDP [https://github.com/JuliaPOMDP/QMDP.jl]

	SARSOP [https://github.com/JuliaPOMDP/SARSOP.jl]

	POMCP [https://github.com/JuliaPOMDP/POMCP.jl]

	DESPOT [https://github.com/JuliaPOMDP/DESPOT.jl]

	MCVI [https://github.com/JuliaPOMDP/MCVI.jl]

	POMDPSolve [https://github.com/JuliaPOMDP/POMDPSolve.jl]






Support Tools:


	POMDPToolbox [https://github.com/JuliaPOMDP/POMDPToolbox.jl]

	POMDPModels [https://github.com/JuliaPOMDP/POMDPModels.jl]








Manual Outline













          

      

      

    

  

    
      
          
            
  
Concepts and Architecture

POMDPs.jl aims to coordinate the development of three software
components: 1) a problem, 2) a solver, 3) an experiment. Each of these
components has a set of abstract types associated with it and a set of
functions that allow a user to define each component’s behavior in a
standardized way. An outline of the architecture is shown below.

[image: concepts]

The MDP and POMDP types are associated with the problem definition. The
Solver and Policy types are associated with the solver or
decision-making agent. Typically, the Updater type is also associated
with the solver, but a solver may sometimes be used with an updater that
was implemented separately. The Simulator type is associated with the
experiment.


POMDPs and MDPs

An MDP is a mathematical framework for sequential decision making under
uncertainty, and where all of the uncertainty arrises from outcomes that
are partially random and partially under the control of a decision
maker. Mathematically, an MDP is a tuple (S,A,T,R), where S is the state
space, A is the action space, T is a transition function defining the
probability of transitioning to each state given the state and action at
the previous time, and R is a reward function mapping every possible
transition (s,a,s’) to a real reward value. For more information see a
textbook such as [1]. In POMDPs.jl an MDP is represented by a concrete
subtype of the MDP abstract type and a set of methods that
define each of its components. S and A are defined by implementing
states and actions for your specific MDP
subtype. R is by implementing reward, and T is defined by implementing transition if the [explicit](@ref defining_pomdps) interface is used or generate_s if the [generative](@ref defining_pomdps) interface is used.

A POMDP is a more general sequential decision making problem in which
the agent is not sure what state they are in. The state is only
partially observable by the decision making agent. Mathematically, a
POMDP is a tuple (S,A,T,R,O,Z) where S, A, T, and R are the same as with
MDPs, Z is the agent’s observation space, and O defines the probability
of receiving each observation at a transition. In POMDPs.jl, a POMDP is
represented by a concrete subtype of the POMDP abstract type,
Z may be defined by the observations function (though an
explicit definition is often not required), and O is defined by
implementing observation if the [explicit](@ref defining_pomdps) interface is used or generate_o if the [generative](@ref defining_pomdps) interface is used.

POMDPs.jl also contains functions for defining optional problem behavior
such as a discount factor or a set of terminal states.

More information can be found in the [Defining POMDPs](@ref defining_pomdps) section.




Beliefs and Updaters

In a POMDP domain, the decision-making agent does not have complete
information about the state of the problem, so the agent can only make
choices based on its “belief” about the state. In the POMDP literature,
the term “belief” is typically defined to mean a probability
distribution over all possible states of the system. However, in
practice, the agent often makes decisions based on an incomplete or
lossy record of past observations that has a structure much different
from a probability distribution. For example, if the agent is
represented by a finite-state controller, as is the case for Monte-Carlo
Value Iteration [2], the belief is the controller state, which is a
node in a graph. Another example is an agent represented by a recurrent
neural network. In this case, the agent’s belief is the state of the
network. In order to accommodate a wide variety of decision-making
approaches in POMDPs.jl, we use the term “belief” to denote the set of
information that the agent makes a decision on, which could be an exact
state distribution, an action-observation history, a set of weighted
particles, or the examples mentioned before. In code, the belief can be
represented by any built-in or user-defined type.

When an action is taken and a new observation is received, the belief is
updated by the belief updater. In code, a belief updater is represented
by a concrete subtype of the Updater abstract type, and the
update function defines how the belief is updated when a new
observation is received.

Although the agent may use a specialized belief structure to make
decisions, the information initially given to the agent about the state
of the problem is usually most conveniently represented as a state
distribution, thus the initialize_belief function is provided
to convert a state distribution to a specialized belief structure that
an updater can work with.

In many cases, the belief structure is closely related to the solution
technique, so it will be implemented by the programmer who writes the
solver. In other cases, the agent can use a variety of belief structures
to make decisions, so a domain-specific updater implemented by the
programmer that wrote the problem description may be appropriate.
Finally, some advanced generic belief updaters such as particle filters
may be implemented by a third party. The convenience function
updater can be used to get a suitable default updater for a
policy, however many policies can work with other updaters.




Solvers and Policies

Sequential decision making under uncertainty involves both online and
offline calculations. In the broad sense, the term “solver” as used in
the node in the figure at the top of the page refers to the software
package that performs the calculations at both of these times. However,
the code is broken up into two pieces, the solver that performs
calculations offline and the policy that performs calculations online.

In the abstract, a policy is a mapping from every belief that an agent
might take to an action. A policy is represented in code by a concrete
subtype of the Policy abstract type. The programmer implements
action to describe what computations need to be done online.
For an online solver such as POMCP, all of the decision computation
occurs within action while for an offline solver like SARSOP,
there is very little computation within action

The offline portion of the computation is carried out by the solver,
which is represented by a concrete subtype of the Solver
abstract type. Computations occur within the solve function.
For an offline solver like SARSOP, nearly all of the decision
computation occurs within this function, but for some online solvers
such as POMCP, solve merely embeds the problem in the policy.




Simulators

A simulator defines a way to run a single simulation. It is represented
by a concrete subtype of the Simulator abstract type and the
simulation is an implemention of simulate. simulate
should return the discounted sum of the stagewise rewards, and the
simulator may or may not keep track of the state trajectory or other
statistics or display the simulation as it is carried out.

[1] Decision Making Under Uncertainty: Theory and Application by
Mykel J. Kochenderfer, MIT Press, 2015

[2] Bai, H., Hsu, D., & Lee, W. S. (2014). Integrated perception and
planning in the continuous space: A POMDP approach. The International
Journal of Robotics Research, 33(9), 1288-1302







          

      

      

    

  

    
      
          
            
  
[Defining POMDPs](@id defining_pomdps)

The expressive nature of POMDPs.jl gives problem writers the flexibility to write their problem in many forms.
Custom POMDP problems are defined by implementing the functions specified by the POMDPs API.


Types of problem definitions

There are two ways of specifying the state dynamics and observation behavior of a POMDP. The problem definition may include either an explicit definition of the probability distributions, or an implicit definition given only by a generative model.

An explicit definition contains the transition probabilities for each state and action, T(s' | s, a), and the observation probabilities for each state-action-state transition, O(o | s, a, s'), (in continuous domains these are probability density functions). A generative definition contains only a single step simulator, s', o, r = G(s, a) (or s', r = G(s,a) for an MDP).

Accordingly, the POMDPs.jl model API is grouped into three sections:


	The [explicit](@ref explicit_api) interface containing functions that return distributions

	The [generative](@ref generative_api) interface containing functions that return states and observations

	[Common](@ref common_api) functions used in both.






What do I need to implement?

Generally, a problem will be defined by implementing either


	An explicit definition consisting of the three functions in (1) and some functions from (3), or

	A generative definition consisting of some functions from (2) and some functions from (3)
(though in some cases (e.g. particle filtering), implementations from all three sections are useful).



Note: since an explicit definition contains all of the information required for a generative definition, POMDPs.jl will automatically generate the generative functions at runtime if an explicit model is available.

An explicit definition will allow for use with the widest variety of tools and solvers, but a generative definition will generally be much easier to implement.

In order to determine which interface to use to express a problem, 2 questions should be asked:


	Is it impossible to specify the probability distributions explicitly (or difficult compared to writing a state simulator)?

	What solvers will be used to solve this, and what are their requirements?



If the answer to (1) is yes, then a generative definition should be used. More information about how to analyze question (2) can be found in the [Requirements](@ref requirements) section of the documentation.

Specific documentation for the two interfaces can be found here:


	[Explicit](@ref explicit_doc)

	[Generative](@ref generative_doc)

julia> using POMDPModels
julia> @requirements_info ValueIterationsolver() GridWorld()

INFO: POMDPs.jl requirements for solve(::ValueIterationSolver, ::Union{POMDP,MDP}) and dependencies. ([v]
= implemented correctly; [X] = missing)

For solve(::ValueIterationSolver, ::Union{POMDP,HDP}):

discount(: :GridWorld)

[v] pdf(::GridWorldpistribution, ::GridWorldstate)

For ordered_states(::Union{POMDP,MDP}) (in solve(::ValueIterationsolver,
[v] states(::GridWorld)

For ordered_actions(::Union{POMDP,MDP}) (in solve(::ValueIterationsolver
[v] action_index(::GridWorld, ::GridWorldAction)
[v] actions(::GridWorld)

julia> type NewMDP <: MDP{Int, Int} end
julia> @requirements_info ValueIterationsolver() NewMDP()

INFO: POMDPs.jl requirements for solve(::ValueIterationsSolver, ::Union{POMDP,MDP}) and dependencies. ([v]
= implemented correctly; [X] = missing)

For solve(::ValueIterationsSolver, ::Union{POMDP,MDP}):
NewMDP)
_: : :NewMDP)
[X] n_actions(
[X] transition(

::Int64, ::Int64)
[X] reward(::NewMDP, ::Int64, ::Int64, ::Int64)
[] state_index(::NewMDP, ::Int64)
WARNING: Some requirements may not be shown because a MethodError was thrown.
For ordered_states(::Union{POMDP,MDP}) (in solve(::ValueIterationSolver, ::Union{POMDP,MDP})):
[X] states(::NewMDP)
WARNING: Some requirements may not be shown because a MethodError was thrown.
For ordered_actions(::Union{POMDP,MDP}) (in solve(::ValueIterationSolver, ::Union{POMDP,MDP})):
[] action_index(::NewMDP, ::Int64)
[X] actions(::NewMDP)
WARNING: Some requirements may not be shown because a MethodError was thrown.

Throwing the first exception (from processing solve(::ValueIterationSolver, ::Union{POMDP,HDP}) requireme
nts):

ERROR: MethodError: no method matching actions(::NewMDP)
Closest candidates are:

actions{S}(::POMDPs.MDP{S,Int64}, ::Range{T}) at /home/zach/.julia/ve.5/POMDPToolbox/src/convenien
ce/implementations.jl:11

actions{s,A}(::Union{POMDPs.MDP{S,A},POMDPS.POMDP{S,A,0}}, ::S) at /home/zach/.julia/ve.5/POMDPs/src/sp
ace.jl:43

actions{s,0}(::POMDPs.POMDP{S,B00l,0}) at /home/zach/.julia/ve.5/POMDPToolbox/src/convenience/implement

ations.jl:6
in macro expansion at /home/zach/.julia/vo.5/Discretevaluelteration/src/vanilla.jl:92 [inlined]

in macro expansion at /home/zach/.julia/ve.5/POMDPs/src/requirements_internals.jl:51 [inlined]

in get_requirements(::POMDPs.#solve, ::Tuple{DiscretevalueIteration.ValueIterationSolver,NewMDP}) at /ho
me/zach/.julia/ve.5/POMDPs/src/requirements_interface.jl:62

in requirements_info(::DiscreteValuelteration.valuelterationSolver,
POMDPs /src/requirements_interface.jl:140

ewMDP) at /home/zach/.julia/ve.5/

